
if (NOT ukui-quick_FIND_COMPONENTS)
    set(ukui-quick_NOT_FOUND_MESSAGE "The ukui-quick package requires at least one component")
    set(ukui-quick_FOUND False)
    return()
endif()

set(_ukui-quick_FIND_PARTS_REQUIRED)
if (ukui-quick_FIND_REQUIRED)
    set(_ukui-quick_FIND_PARTS_REQUIRED REQUIRED)
endif()
set(_ukui-quick_FIND_PARTS_QUIET)
if (ukui-quick_FIND_QUIETLY)
    set(_ukui-quick_FIND_PARTS_QUIET QUIET)
endif()

get_filename_component(_ukui-quick_install_prefix "${CMAKE_CURRENT_LIST_DIR}/.." ABSOLUTE)

set(_ukui-quick_root_dir ${_ukui-quick_install_prefix})
set(_ukui-quick_module_paths ${_ukui-quick_install_prefix})

set(_at @)
set(_ukui-quick_module_location_template ${_ukui-quick_install_prefix}/ukui-quick-${_at}module${_at}/ukui-quick-${_at}module${_at}-config.cmake)
unset(_at)
set(_ukui-quick_NOTFOUND_MESSAGE)

foreach(module ${ukui-quick_FIND_COMPONENTS})
    find_package(ukui-quick-${module}
        ${_ukui-quick_FIND_PARTS_QUIET}
        ${_ukui-quick_FIND_PARTS_REQUIRED}
        PATHS ${_ukui-quick_module_paths} NO_DEFAULT_PATH
    )
    if (NOT ukui-quick-${module}_FOUND)
        string(CONFIGURE ${_ukui-quick_module_location_template} _expected_module_location @ONLY)

        if (ukui-quick_FIND_REQUIRED_${module})
            set(_ukui-quick_NOTFOUND_MESSAGE "${_ukui-quick_NOTFOUND_MESSAGE}Failed to find ukui-quick component \"${module}\" config file at \"${_expected_module_location}\"\n")
        elseif(NOT ukui-quick_FIND_QUIETLY)
            message(WARNING "Failed to find ukui-quick component \"${module}\" config file at \"${_expected_module_location}\"")
        endif()

        unset(_expected_module_location)
    endif()
endforeach()
if (_ukui-quick_NOTFOUND_MESSAGE)
    set(ukui-quick_NOT_FOUND_MESSAGE "${_ukui-quick_NOTFOUND_MESSAGE}")
    set(ukui-quick_FOUND False)
endif()